Techniques d'intégration *(partie 1 de 5)*

Changement de variables

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2003. L'objectif principal de cette feuille Maple est de rendre l'élève apte à transposer en Maple la technique de changement de variables. Cette transposition sera réalisée avec certaines macro-commandes de la sous-bibliothèque Calculus1 de la bibliothèque Student.

Les exemples développés dans ce document illustreront la manière d'obtenir l'intégrale indéfinie de deux façons:

- une première façon consistera à obtenir directement l'intégrale indéfinie avec la macro-commande int de la bibliothèque principale, question d'observer le mécanisme de la simplification automatique de Maple.
- la seconde façon, quant à elle, consistera à faire un développement pas à pas tel un développement *manus scriptus*. Un tel développement sera réalisé selon les règles d'intégration Rule de *Student*[*Calculus1*].

Pour compléter tout développement pas à pas, la dernière étape consistera à faire une vérification de la réponse finale par dérivation.

Remarque: L'exemple 5 nous donnera l'occasion d'une discussion sur les modèles $\int \frac{1}{u\sqrt{u^2-a^2}}\,\mathrm{d}x$ et $\int \frac{1}{|u|\sqrt{u^2-a^2}}\,\mathrm{d}x$.

Attention:

- S'assurer que la rubrique « Afficher les étiquettes des équations » soit cochée: voir menu Outils → Options… → Affichage.
- S'assurer que le préfixe de numérotation d'étiquettes soit initialisée à 1 avec un schéma de numérotation séquentielle: voir menu Format → Étiquettes d'équation (L) → Affichage des étiquettes… (L).



- L'insertion d'une étiquette est obtenu avec le raccourci « ctrl + l » (l pour label en anglais).

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
```

Assurance que les étiquettes des équations soient affichées.

```
> interface(labelling=true)
```

$$true \qquad (1)$$

Réglage de l'affichage pour les variables conditionnées.

```
> with(Typesetting):
  interface(typesetting=standard); #niveau de composition Maple
  Standard
  interface(showassumed=2);         # Variables avec suppositiosnOhrase
```

$$extended,\ [extended]$$

$$2 \qquad (2)$$

Intégration par changement de variables

Une primitive d'une fonction f est une fonction F dont la dérivée $\mathrm{F}'(x) = \mathrm{f}(x)$. La recherche d'une primitive est, en général, assez délicate. Contrairement à l'opération de dérivation, l'opération de primitivation repose davantage sur des techniques plutôt que sur des règles simples de calcul comme c'est le cas pour la dérivation ou pour le calcul de limites. C'est par l'apprentissage d'un ensemble de techniques qui permettra à l'élève d'acquérir une expérience lui permettant de développer une intuition pertinente pour effectuer cette opération. Voici donc une toute première technique de base.

La technique d'intégration présentée au cours de cette feuille Maple est la technique de base de changement de variables. C'est une technique d'intégration qui découle de la règle de dérivation en chaîne. Soit deux fonctions f et g. Rappelons que la dérivée de $(\mathrm{fog})(x)$ est donnée par

$$\frac{\mathrm{d}}{\mathrm{d}x}(\mathrm{fog})(x) = \left(\frac{\mathrm{d}}{\mathrm{d}[\mathrm{g}(x)]}\mathrm{f}[\mathrm{g}(x)]\right)\left(\frac{\mathrm{d}}{\mathrm{d}x}\mathrm{g}(x)\right) \qquad (1)$$

Sachant que $\int \mathrm{f}'(x)\,\mathrm{d}x = \int \mathrm{d}(\mathrm{f}(x)) = \mathrm{f}(x) + C$, l'égalité (1) permet de poser

$$\int\left(\frac{\mathrm{d}}{\mathrm{d}[\mathrm{g}(x)]}\mathrm{f}[\mathrm{g}(x)]\right)\left(\frac{\mathrm{d}}{\mathrm{d}x}\mathrm{g}(x)\right)\mathrm{d}x = \int \mathrm{d}((\mathrm{fog})(x)) = (\mathrm{fog})(x) + \mathrm{C}$$

Illustrons ces égalités précédentes avec l'exemple suivant. Intégrons la fonction h définie par $\mathrm{h}(x) = 2x\cos x^2$. Peut-on trouver une fonction H dont la dérivée soit la fonction h? En appliquant la règle de la dérivation en chaîne, on observe que

$$\frac{\mathrm{d}\sin(x^2)}{\mathrm{d}x} = 2x\cos x^2 = \mathrm{h}(x).$$

Ainsi,

$$\int \mathrm{h}(x)\,\mathrm{d}x = \int 2x\cos(x^2)\,\mathrm{d}x = \int d(\sin(x^2)) = \sin x^2 + C.$$

Or, de manière pratico-pratique, à la lecture de l'intégrale indéfinie

$$\int 2x\cos(x^2)\,\mathrm{d}x$$

il faut être capable de (intuitionner pour) reconnaître que l'intégrande est le résultat de l'application de la règle de dérivation en chaîne qui a été appliquée à la composition $\mathrm{f}(u) = \sin(u)$ où $u = \mathrm{g}(x) = x^2$. Le moyen qu'on se donne pour s'aider à visualiser la dérivée d'une composition de deux fonctions est d'effectuer un changement de variables. Dans ce cas-ci, posons la substitution $u = \mathrm{g}(x) = x^2$ pour transformer la différentielle $2x\cos x^2\,dx$ en la différentielle $\cos(u)\,du$. En effet, Si on pose $u = x^2$

cela entraîne que $\mathrm{d}u = 2\,x\,\mathrm{d}x$

De cette manière, on peut déduire les égalités suivantes

$$\int 2\,x\cos\left(x^2\right)\,\mathrm{d}x = \int \cos\left(x^2\right)\cdot 2\,x\,\mathrm{d}x = \int \cos(u)\,\mathrm{d}u = \sin(u) + C = \sin\left(x^2\right) + C.$$

La technique d'intégration par changement de variables ne peut pas, malheureusement, s'appliquer à n'importe différentielle à intégrer. Cette technique sera pertinente seulement si on reconnaît au départ que l'intégrande est le résultat de la dérivation d'une composition de deux fonctions (à une constante multiplicatrice près).

En plus d'aborder la technique d'intégration de base par changement de variables, cette feuille Maple sera aussi l'occasion d'utiliser un nombre accru de macro-commandes dédiées à la simplification sur demande.

Rappelons que nous allons obtenir l'intégrale indéfinie de deux façons. Une première façon en utilisant directement la macro-commande int, question d'observer le mécanisme de la simplification automatique. La seconde, avec un développement pas à pas (manus scriptus) en appliquant les règles de l'extension *Calculus1* jusqu'à l'obtention d'un des modèles de base retenus en classe. À cette ultime étape du développement pas à pas, on utililisera la macro-commande *value* pour appliquer simultanément le modèle de base et la substitution à rebours.

Exemple 1

Calculons l'intégrale indéfinie $\int (5\,x-3)\sqrt{2-7\,x}\,\mathrm{d}x$.

Question d'observer le mécanisme de la simplification automatique, intégrons directement $\int (5\,x-3)\sqrt{2-7\,x}\,\mathrm{d}x$ avec la macro-commande *int***.** Afin de bien documenter les prochains développements, créons d'abord une fonction f (avec l'opérateur flèche) dont la règle f(*x*) sera l'intégrande.

```
> f:=x->(5*x-3)*sqrt(2-7*x);
```

$$f := x \rightarrow (5\,x-3)\sqrt{2-7\,x} \qquad \textbf{(3)}$$

Pour que l'intégrale indéfinie puisse s'afficher avec une notation mathématique habituelle dans la zone des résultats, utilisons la forme inactive de la macro-commande *int.* De plus, donnons le nom *Problème* à l'intégrale indéfinie.

```
> Problème:=Int(f(x),x);
```

$$\mathit{Problème} := \int (5\,x-3)\sqrt{2-7\,x}\,\mathrm{d}x \qquad \textbf{(4)}$$

Noter, dans la zone des résultats, la manière dont l'afficheur utilise le gris pour afficher les caractères « $\int$ » et « d ». C'est de cette manière que Maple signifie clairement que la mise en forme du résultat est posée expressément par l'utilisateur et n'est donc pas le résultat d'une évaluation (manquée) par Maple.

Obtenons maintenant cette intégrale indéfinie directement avec la macro-commande *value.* Assignons le résultat au nom Primitive.

```
> Primitive:=value(Problème);
Error, attempting to assign to `Primitive` which is protected.  Try declaring `local Primitive`; see ?protect for details.
```

Le message d'erreur est très clair: le nom *Primitive* est un nom réservé par Maple. C'est en fait une macro-commande de la bibliothèque de base. Modifions donc nos habitudes. Au lieu de commencer le nom choisi par une lettre majuscule, commençons-le plutôt avec une lettre minuscule car nous tenons ici à conserver le nom significatif « Primitive ».

```
> primitive:=value(Problème);
```

$$primitive := -\frac{2}{147}\,(21\,x-17)\,(2-7\,x)^{3/2} \qquad \textbf{(5)}$$

Remarque: Le résultat est donné sans constante d'intégration (disons plutôt que les programmeurs de Maple ont choisi, par commodité, 0 comme constante d'intégration). *Il vous faudra donc gérer manuellement la globalisation des constantes d'intégration dans votre réponse finale.* Pour obtenir l'ensemble de toutes les primitives tel que demandé par l'intégrale indéfinie, ajoutons ponctuellement une constante *C* dans la formulation de la réponse finale.

```
> Rép_Maple:=Problème=primitive+C;
```

$$Rép\_Maple := \int (5\,x-3)\sqrt{2-7\,x}\,dx = -\frac{2}{147}\,(21\,x-17)\,(2-7\,x)^{3/2}+C \qquad \textbf{(6)}$$

La plupart du temps nous laisserons tel quel le résultat. Par contre, dans le second développement *manus scriptus*, nous serons attentif à une plus grande simplification pour la formulation de la réponse finale. Par exemple, s'il y a lieu, on effectuera les mises en évidence qui s'imposent.

Obtenons maintenant $\int (5\,x-3)\sqrt{2-7\,x}\,dx$ avec un développement pas à pas. La technique de changement de variables consiste à transformer la différentielle de départ en une différentielle correspondant à un modèle de base retenu en classe dont la primitive est connue par cœur.

L'extension Student[Calculus1][Int] sera utilisée afin de transposer en Maple les différentes techniques d'intégration vues en classe. Plusieurs macro-commandes de cette extension seront utiles afin de développer pas à pas une intégrale indéfinie jusqu'à ce que cela puisse permettre de reconnaître l'un des dix-sept modèles de base.

Effectuons le changement de variables en posant $u = 2-7\,x$. Pour ce faire, rendons actives les macro-commandes *Rule* et *Integrand* de *Student*[*Calculus1*]. La macro-commande *Rule* servira à appliquer les propriétés de l'intégrale indéfinie ainsi que les modèles de base tandis que la macro-commande *Integrand* nous permettera de pointer vers l'intégrande de l'intégrale.

```
> with(Student[Calculus1],Rule,Integrand);
  infolevel[Student[Calculus1]]:=1:
```

$$[Rule, Integrand] \qquad \textbf{(7)}$$

Attention: Soyez attentif à la manière dont est documenté le développement pas à pas qui transpose en Maple la technique d'intégration de base par changement de variables. Ce sera le modèle de développement à suivre pour développer les intégrales indéfinies de votre TP.

Rappelons-nous l'énoncé de l'intégrale indéfinie à calculer.

```
> Problème;
```

$$\int (5\,x-3)\sqrt{2-7\,x}\,dx \qquad \textbf{(8)}$$

Effectuons le changement de variables suivant: posons $u = 2-7\,x$.

```
> Rule[change,u=2-7*x]((8));
Creating problem #1

Applying substitution x = 2/7-1/7*u, u = 2-7*x with dx = -1/7*du, du =
-7*dx
```

$$\int (5\,x-3)\sqrt{2-7\,x}\,dx = \int \frac{1}{49}\sqrt{u}\,(11+5\,u)\,du \qquad \textbf{(9)}$$

Notons que la macro-commande $Rule_{change}$ a transformé automatiquement la différentielle de départ en la reformulant complètement en termes de la nouvelle variable choisie, c'est-à-dire en termes de la variable u dans ce cas-ci. En conséquence, la substitution de la différentielle dx a été automatiquement pris en charge par la macro-commande.

Appliquons la règle de la constante multiplicatrice.

```
> Rule[`c*`]((9));
```

$$\int (5x-3)\sqrt{2-7x}\,\mathrm{d}x = \frac{1}{49}\int \sqrt{u}\,(11+5u)\,\mathrm{d}u \tag{10}$$

Avant d'appliquer la règle $Rule_{sum}$, nous devons d'abord réécrire l'intégrande comme une somme. Il est évident que nous n'avons pas besoin de Maple pour réécrire $\sqrt{u}\,(11+5u)$ par $11\,u^{\frac{1}{2}}+5\,u^{\frac{3}{2}}$. Voyons tout de même comment s'y prendre avec Maple.

```
> Integrand(rhs((10)));
```

$$\left[\sqrt{u}\,(11+5u)\right] \tag{11}$$

```
> expand((11));
```

$$\left[11\sqrt{u}+5\,u^{3/2}\right] \tag{12}$$

Pointons vers l'intégrande.

```
> Rule[rewrite,sqrt(u)*(11+5*u)=(12)[1]]((10));
```

$$\int (5x-3)\sqrt{2-7x}\,\mathrm{d}x = \frac{1}{49}\int \left(11\sqrt{u}+5\,u^{3/2}\right)\mathrm{d}u \tag{13}$$

Appliquons la règle de linéarité.

```
> Rule[sum]((13));
```

$$\int (5x-3)\sqrt{2-7x}\,\mathrm{d}x = \frac{1}{49}\int 11\sqrt{u}\,\mathrm{d}u + \frac{1}{49}\int 5\,u^{3/2}\,\mathrm{d}u \tag{14}$$

Appliquons la règle de la constante multiplicatrice.

```
> Rule[`c*`]((14));
```

$$\int (5x-3)\sqrt{2-7x}\,\mathrm{d}x = \frac{11}{49}\int \sqrt{u}\,\mathrm{d}u + \frac{1}{49}\int 5\,u^{3/2}\,\mathrm{d}u \tag{15}$$

Appliquons encore une fois cette règle.

```
> Rule[`c*`]((15));
```

$$\int (5x-3)\sqrt{2-7x}\,\mathrm{d}x = \frac{11}{49}\int \sqrt{u}\,\mathrm{d}u + \frac{5}{49}\int u^{3/2}\,\mathrm{d}u \tag{16}$$

Notre but est atteint: les deux intégrales indéfinies précédentes correspondent chacune à un modèle de base (dans ce cas-ci, c'est un même modèle):

$$\int u^n\,\mathrm{d}u = \frac{u^{n+1}}{n+1} + C \qquad \text{pour } n \neq -1:$$

```
> Rule[power]((16));
```

$$\int (5x-3)\sqrt{2-7x}\,\mathrm{d}x = \frac{11}{49}\int \sqrt{u}\,\mathrm{d}u + \frac{2}{49}\,u^{5/2} \tag{17}$$

```
> Rule[power]((17));
```

$$\int (5x-3)\sqrt{2-7x}\,dx = \frac{22}{147}u^{3/2} + \frac{2}{49}u^{5/2} \quad (18)$$

Dans une zone de résultats, formulons finalement l'intégrale indéfinie $\int (5x-3)\sqrt{2-7x}\,dx$.

```
> factor((18));
```

$$\int (5x-3)\sqrt{2-7x}\,dx = \frac{2}{147}u^{3/2}(3u+11) \quad (19)$$

Effectuons la substituions à rebours

```
> Rép_finale:=Rule[revert]((19))+(0=C);
Reverting substitution using u = 2-7*x
```

$$Rép\_finale := \int (5x-3)\sqrt{2-7x}\,dx = \frac{2}{147}(2-7x)^{3/2}(-21x+17) + C \quad (20)$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(\frac{2}{147}(2-7x)^{3/2}(-21x+17)+C\right) = -\frac{1}{7}\sqrt{2-7x}\,(-21x+17) - \frac{2}{7}(2-7x)^{3/2} \quad (21)$$

Simplifions le membre de droite en factorisant $\sqrt{2-7x}$.

```
> ``=factor(rhs((21)));
```

$$= (5x-3)\sqrt{2-7x} \quad (22)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$(5x-3)\sqrt{2-7x} \quad (23)$$

Exemple 2

Calculons l'intégrale indéfinie $\int 2x^2\left(5-x^3\right)^8 dx$.

D'abord directement avec Maple.

```
> f:=x->2*x^2*(5-x^3)^8;
```

$$f := x \to 2x^2\left(5-x^3\right)^8 \quad (24)$$

Posons l'intégrale indéfinie $\int 2x^2\left(5-x^3\right)^8 dx$ qu'il faut obtenir.

```
> Problème:=Int(f(x),x);
```

$$Problème := \int 2x^2\left(-x^3+5\right)^8 dx \quad (25)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande *value*.

```
> primitive:=value(Problème);
```

$$primitive := \frac{2}{27}x^{27} - \frac{10}{3}x^{24} + \frac{200}{3}x^{21} - \frac{7000}{9}x^{18} + \frac{17500}{3}x^{15} - \frac{87500}{3}x^{12} + \frac{875000}{9}x^{9} - \frac{625000}{3}x^{6} + \frac{781250}{3}x^{3} \quad (26)$$

```
> Rép_Maple:=Problème=primitive+C1;
```

$$Rép\_Maple := \int 2x^2\left(-x^3+5\right)^8 dx = \frac{2}{27}x^{27} - \frac{10}{3}x^{24} + \frac{200}{3}x^{21} - \frac{7000}{9}x^{18} + \frac{17500}{3}x^{15} - \frac{87500}{3}x^{12} + \frac{875000}{9}x^9 - \frac{625000}{3}x^6 + \frac{781250}{3}x^3 + C1 \qquad (27)$$

Intégrons maintenant avec un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie qu'on cherche à obtenir.

```
> Problème;
```

$$\int 2x^2\left(-x^3+5\right)^8 dx \qquad (28)$$

Posons $u = 5 - x^3$.

```
> Rule[change,u=5-x^3]((28));
Creating problem #2

Applying substitution x = (5-u)^(1/3), u = -x^3+5 with dx = -1/3/(5-u)^
(2/3)*du, du = -3*x^2*dx
```

$$\int 2x^2\left(-x^3+5\right)^8 dx = \int\left(-\frac{2}{3}u^8\right) du \qquad (29)$$

Sortons la constante multiplicatrice $-\frac{2}{3}$ de l'intégrale indéfinie.

```
> Rule[`c*`]((29));
```

$$\int 2x^2\left(-x^3+5\right)^8 dx = -\frac{2}{3}\int u^8\, du \qquad (30)$$

Notre but est atteint, on reconnaît un modèle de base.

```
> Rule[power]((30));
```

$$\int 2x^2\left(-x^3+5\right)^8 dx = -\frac{2}{27}u^9 \qquad (31)$$

```
> Rép_finale:=Rule[revert](((31)))+(0=C);
Reverting substitution using u = -x^3+5
```

$$Rép\_finale := \int 2x^2\left(-x^3+5\right)^8 dx = \frac{2}{27}x^{27} - \frac{10}{3}x^{24} + \frac{200}{3}x^{21} - \frac{7000}{9}x^{18} + \frac{17500}{3}x^{15} - \frac{87500}{3}x^{12} + \frac{875000}{9}x^9 - \frac{625000}{3}x^6 + \frac{781250}{3}x^3 - \frac{3906250}{27} + C \qquad (32)$$

La susbtituion à rebours n'a pas donné le résultat attendu $-\frac{2\left(5-x^3\right)^9}{27}$ mais plutôt la forme développée.

```
> expand(-2*(-x^3+5)^9*(1/27));
```

$$\frac{2}{27}x^{27} - \frac{10}{3}x^{24} + \frac{200}{3}x^{21} - \frac{7000}{9}x^{18} + \frac{17500}{3}x^{15} - \frac{87500}{3}x^{12} + \frac{875000}{9}x^9 \qquad (33)$$

$$-\frac{625000}{3}x^6+\frac{781250}{3}x^3-\frac{3906250}{27}$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(\frac{2}{27}x^{27}-\frac{10}{3}x^{24}+\frac{200}{3}x^{21}-\frac{7000}{9}x^{18}+\frac{17500}{3}x^{15}-\frac{87500}{3}x^{12}+\frac{875000}{9}x^9-\frac{625000}{3}x^6+\frac{781250}{3}x^3-\frac{3906250}{27}+C\right)=2x^{26}-80x^{23}+1400x^{20}-14000x^{17}+87500x^{14}-350000x^{11}+875000x^8-1250000x^5+781250x^2 \quad \textbf{(34)}$$

```
> Diff(rhs(Rép_finale),x)=factor(diff(rhs(Rép_finale),x));
```

$$\frac{\partial}{\partial x}\left(\frac{2}{27}x^{27}-\frac{10}{3}x^{24}+\frac{200}{3}x^{21}-\frac{7000}{9}x^{18}+\frac{17500}{3}x^{15}-\frac{87500}{3}x^{12}+\frac{875000}{9}x^9-\frac{625000}{3}x^6+\frac{781250}{3}x^3-\frac{3906250}{27}+C\right)=2x^2\left(x^3-5\right)^8 \quad \textbf{(35)}$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$2x^2\left(-x^3+5\right)^8 \quad \textbf{(36)}$$

Montrons finalement que le résultat de la simplification automatique et celui obtenu avec un développement pas à pas ne diffèrent que par une constante.

```
> rhs(Rép_Maple)-rhs(Rép_finale);
```

$$C1+\frac{3906250}{27}-C \quad \textbf{(37)}$$

Exemple 3

Calculons l'intégrale indéfinie $\int\left(3x^4+3\right)\left(\sec^2\right)\left(x^5+5x\right)\,dx$.

D'abord directement avec Maple.

```
> f:=x->(3*x^4+3)*(sec^2)(x^5+5*x);
```

$$f:=x\to\left(3x^4+3\right)\left(\sec^2\right)\left(x^5+5x\right) \quad \textbf{(38)}$$

```
> Problème:=Int(f(x),x);
```

$$Problème:=\int\left(3x^4+3\right)\sec\left(x^5+5x\right)^2\,dx \quad \textbf{(39)}$$

```
> primitive:=value(Problème);
```

$$primitive:=\frac{3}{5}\tan\left(x^5+5x\right) \quad \textbf{(40)}$$

```
> Rép_Maple:=Problème=primitive+C;
```

(41)

$$Rép\_Maple := \int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \frac{3}{5}\tan\left(x^5+5x\right)+C \tag{41}$$

Intégrons maintenant en effectuant un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie qu'on cherche à obtenir.

```
> Problème;
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x \tag{42}$$

Posons $u = x^5 + 5x$.

```
> Rule[change,u=x^5+5*x]((42));
Creating problem #3

Applying substitution x = RootOf(_Z^5+5*_Z-u), u = x^5+5*x with dx = 1/
(5*RootOf(_Z^5+5*_Z-u)^4+5)*du, du = (5*x^4+5)*dx
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \int \frac{3}{5}\sec(u)^2 \mathrm{d}u \tag{43}$$

```
> Rule[`c*`]((43));
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \frac{3}{5}\int \sec(u)^2 \mathrm{d}u \tag{44}$$

Notre but est atteint, le modèle de base $\int \left(\sec^2\right)(x)dx = \tan(x)+C$…mais quant même, complétons pas à pas le développement.

```
> Rule[change,t=tan(u)]((44));
Applying substitution u = arctan(t), t = tan(u) with du = 1/(t^2+1)*dt,
dt = (1+tan(u)^2)*du
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \frac{3}{5}\int 1 \,\mathrm{d}t \tag{45}$$

```
> Rule[constant]((45));
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \frac{3}{5}\,t \tag{46}$$

```
> Rule[revert]((46));
Reverting substitution using t = tan(u)
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \frac{3}{5}\tan(u) \tag{47}$$

```
> Rule[revert]((47));
Reverting substitution using u = x^5+5*x
```

$$\int \left(3x^4+3\right)\sec\left(x^5+5x\right)^2 \mathrm{d}x = \frac{3}{5}\tan\left(x^5+5x\right) \tag{48}$$

```
> Rép_finale:=(48)+(0=C);
```

$$Rép\_finale := \int (3x^4+3)\sec(x^5+5x)^2\,dx = \frac{3}{5}\tan(x^5+5x)+C \qquad (49)$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(\frac{3}{5}\tan(x^5+5x)+C\right) = \frac{3}{5}(5x^4+5)\left(1+\tan(x^5+5x)^2\right) \qquad (50)$$

Simplifions en substituant $(\sec^2)(x^5+5x)$ à $1+(\tan^2)(x^5+5x)$ dans le précédent résultat.

```
> ``=subs(1+(tan^2)(x^5+5*x) = (sec^2)(x^5+5*x),rhs((50)));
```

$$= \frac{3}{5}(5x^4+5)\sec(x^5+5x)^2 \qquad (51)$$

```
> ``=normal(rhs((51)));
```

$$= 3(x^4+1)\sec(x^5+5x)^2 \qquad (52)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$(3x^4+3)\sec(x^5+5x)^2 \qquad (53)$$

Exemple 4

Calculons l'intégrale indéfinie $\int x^3\sqrt{1-x^2}\,dx$.

D'abord directement avec Maple.

```
> f:=x->x^3*sqrt(1-x^2);
```

$$f := x \rightarrow x^3\sqrt{1-x^2} \qquad (54)$$

```
> Problème:=Int(f(x),x);
```

$$Problème := \int x^3\sqrt{-x^2+1}\,dx \qquad (55)$$

```
> primitive:=value(Problème);
```

$$primitive := \frac{1}{15}(x-1)(x+1)(3x^2+2)\sqrt{-x^2+1} \qquad (56)$$

```
> Rép_Maple:=Problème=primitive+C;
```

$$Rép\_Maple := \int x^3\sqrt{-x^2+1}\,dx = \frac{1}{15}(x-1)(x+1)(3x^2+2)\sqrt{-x^2+1}+C \qquad (57)$$

Intégrons maintenant en effectuant un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'intégrale indéfinie à obtenir.

```
> Problème;
```

$$\int x^3\sqrt{-x^2+1}\,dx \qquad (58)$$

Posons $u = 1 - x^2$.

```
> Rule[change,u=1-x^2]((58));
Creating problem #4

Applying substitution x = (-u+1)^(1/2), u = -x^2+1 with dx = -1/2/(-u+1)
^(1/2)*du, du = -2*x*dx
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \int \frac{1}{2}\sqrt{u}\,(u-1)\,\mathrm{d}u \tag{59}$$

Développons à l'aide des propriétés de l'intégrale indéfinie.

```
> expand((59));
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \frac{1}{2}\int\left(u^{3/2}-\sqrt{u}\right)\mathrm{d}u \tag{60}$$

```
> Rule[sum]((60));
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \frac{1}{2}\int u^{3/2}\,\mathrm{d}u + \frac{1}{2}\int\left(-\sqrt{u}\right)\mathrm{d}u \tag{61}$$

```
> Rule[`c*`]((61));
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \frac{1}{2}\int u^{3/2}\,\mathrm{d}u - \frac{1}{2}\int\sqrt{u}\,\mathrm{d}u \tag{62}$$

```
> Rule[power]((62));
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \frac{1}{2}\int u^{3/2}\,\mathrm{d}u - \frac{1}{3}u^{3/2} \tag{63}$$

```
> Rule[power]((63));
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \frac{1}{5}u^{5/2} - \frac{1}{3}u^{3/2} \tag{64}$$

```
> Rule[revert]((64));
Reverting substitution using u = -x^2+1
```

$$\int x^3\sqrt{-x^2+1}\,\mathrm{d}x = \frac{1}{5}\left(-x^2+1\right)^{5/2} - \frac{1}{3}\left(-x^2+1\right)^{3/2} \tag{65}$$

```
> Rép_finale:=factor((65))+(0=C);
```

$$Rép\_finale := \int x^3\sqrt{-x^2+1}\,\mathrm{d}x = -\frac{1}{15}\left(-(x-1)(x+1)\right)^{3/2}\left(3x^2+2\right)+C \tag{66}$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(-\frac{1}{15}\left(-(x-1)(x+1)\right)^{3/2}\left(3x^2+2\right)+C\right) = \frac{1}{5}\sqrt{-(x-1)(x+1)}\,\left(3x^2+2\right)x \tag{67}$$

$$-\frac{2}{5}\left(-(x-1)(x+1)\right)^{3/2}x$$

```
> ``=radnormal(rhs((67)),rationalized);
```

$$=\sqrt{-(x-1)(x+1)}\,x^3 \qquad (68)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$x^3\sqrt{-x^2+1} \qquad (69)$$

```
> factor((69));
```

$$\sqrt{-(x-1)(x+1)}\,x^3 \qquad (70)$$

Exemple 5

Calculons l'intégrale indéfinie $\int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,dx$.

Remarque: Cet exemple nous conduit à une discussion sur les modèles $\int \frac{1}{u\sqrt{u^2-a^2}}\,dx$ et $\int \frac{1}{|u|\sqrt{u^2-a^2}}\,dx$.

D'abord directement avec Maple.

```
> f:=x->1/(x*ln(x)*sqrt((ln(x))^2-1));
```

$$f := x \to \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}} \qquad (71)$$

Posons l'intégrale indéfinie $\int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,dx$ à obtenir.

```
> Problème:=Int(f(x),x);
```

$$Problème := \int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,dx \qquad (72)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande `value`.

```
> primitive:=value(Problème);
```

$$primitive := -\arctan\left(\frac{1}{\sqrt{\ln(x)^2-1}}\right) \qquad (73)$$

```
> Rép_Maple:=Problème=primitive+C;
```

$$Rép\_Maple := \int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,dx = -\arctan\left(\frac{1}{\sqrt{\ln(x)^2-1}}\right)+C \qquad (74)$$

Notons que la simplification automatique n'a donnée ni l'un ou l'autre des modèles attendus.

Intégrons maintenant en effectuant un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie qu'on cherche à obtenir.

```
> Problème;
```

$$\int \frac{1}{x \ln(x) \sqrt{\ln(x)^2 - 1}} \, dx \qquad (75)$$

Posons $u = \ln(x)$.

```
> Rule[change,u=ln(x)]((75));
Creating problem #5

Applying substitution x = exp(u), u = ln(x) with dx = exp(u)*du, du =
1/x*dx
```

$$\int \frac{1}{x \ln(x) \sqrt{\ln(x)^2 - 1}} \, dx = \int \frac{1}{u \sqrt{u^2 - 1}} \, du \qquad (76)$$

Notre but est atteint, on reconnaît un des deux modèle de base, celui de l'arcsécante. Obtenons maintenant cette intégrale indéfinie.

```
> Rule[change,u=sec(t)]((76));
Applying substitution u = sec(t), t = arcsec(u) with du = sec(t)*tan(t)*
dt, dt = 1/u^2/(1-1/u^2)^(1/2)*du
```

$$\int \frac{1}{x \ln(x) \sqrt{\ln(x)^2 - 1}} \, dx = \int 1 \, dt \qquad (77)$$

```
> Rule[constant]((77));
```

$$\int \frac{1}{x \ln(x) \sqrt{\ln(x)^2 - 1}} \, dx = t \qquad (78)$$

```
> Rule[revert]((78));
Reverting substitution using t = arcsec(u)
```

$$\int \frac{1}{x \ln(x) \sqrt{\ln(x)^2 - 1}} \, dx = \operatorname{arcsec}(u) \qquad (79)$$

```
> Rule[revert]((79));
Reverting substitution using u = ln(x)
```

$$\int \frac{1}{x \ln(x) \sqrt{\ln(x)^2 - 1}} \, dx = \operatorname{arcsec}(\ln(x)) \qquad (80)$$

```
> Rép_finale:=(80)+(0=C);
```

(81)

$$Rép\_finale := \int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,dx = \operatorname{arcsec}(\ln(x)) + C \qquad \textbf{(81)}$$

Oups! On s'attendait plutôt avoir comme résultat $\operatorname{arcsec}(|\ln(x)|)$, conformément au modèle retenu en classe. Mais, ce n'est pas ce modèle qui a été retenu par le logiciel. Acceptons, dans un premier temps, ce résultat pour terminer le développement pas à pas. Puis, dans un second temps, nous allons reprendre ce développement en affectant ponctuellement le modèle de base $\operatorname{arcsec}(|\ln(x)|)$ (Voir la remarque plus bas) .

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}(\operatorname{arcsec}(\ln(x)) + C) = \frac{1}{x\ln(x)^2\sqrt{1-\frac{1}{\ln(x)^2}}} \qquad \textbf{(82)}$$

```
> ``=normal(rhs((82)));
```

$$= \frac{1}{x\ln(x)^2\sqrt{\frac{\ln(x)^2-1}{\ln(x)^2}}} \qquad \textbf{(83)}$$

```
> ``=simplify(rhs((83))) assuming x>1
```

$$= \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}} \qquad \textbf{(84)}$$

Cette dernière expression est-elle égale à l'intégrande de Problème ?

```
> Integrand(Problème)[1];
```

$$\frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}} \qquad \textbf{(85)}$$

Remarque: Dans des développements pas à pas que vous aurez à faire pour votre rapport de laboratoire, lorsqu'une situation semblable se produira, vous devrez effectivement assigner ponctuellement le modèle de base retenu en classe avant de poursuivre votre développement.

****
******************************** REPRISE AVEC LA FORMULE DE BASE *************************
****

Reprenons ce développement pas à pas en affectant ponctuellement $\operatorname{arcsec}(|u|)$ à **(76)** conformément au modèle qui a été retenu en classe.

```
> Autre_réponse:=Problème=arcsec(abs(u));
```

$$Autre\_réponse := \int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,dx = \operatorname{arcsec}(|u|) \qquad \textbf{(86)}$$

Ayant posé $u = \ln(x)$, effectuons, à rebours, ce changement de variables dans **(86)**.

```
> ``=subs(u=ln(x),rhs((86)));
```

$$= \operatorname{arcsec}(|\ln(x)|) \tag{87}$$

```
> Autre_réponse_finale:=Problème=rhs((87))+C;
```

$$Autre\_réponse\_finale := \int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,\mathrm{d}x = \operatorname{arcsec}(|\ln(x)|) + C \tag{88}$$

Vérification de la réponse finale par dérivation.

```
> Éq:=Diff(lhs(Autre_réponse_finale),x)=diff(rhs
  (Autre_réponse_finale),x);
```

$$Éq := \frac{\mathrm{d}}{\mathrm{d}x}\left(\int \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}\,\mathrm{d}x\right) = \frac{\operatorname{abs}(1,\ln(x))}{x|\ln(x)|^2\sqrt{1-\frac{1}{|\ln(x)|^2}}} \tag{89}$$

On simplifiera donc explicitement la dérivée en tenant compte du domaine de définition de l'intégrande.

Pour que l'intégrande $\frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}}$ soit définie, nous devons tenir compte de:

1. la présence de x au dénominateur. Cela impose la restriction $x \neq 0$.
2. la présence de $\ln(x)$ au dénominateur. Cela impose les restrictions $x > 0$ et $x \neq 1$.
3. la présence de $\ln(x)^2 - 1$ sous le radicale. Cela impose la restriction $\ln(x)^2 > 1$ $|\ln(x)| > 1$.

```
> solve(abs(ln(x))>1);
```

$$RealRange(Open(\mathrm{e}), \infty), RealRange\left(Open(0), Open\left(\frac{1}{\mathrm{e}}\right)\right) \tag{90}$$

On déduit donc que le domaine de définition de l'intégrande est $\left]0, \frac{1}{\mathrm{e}}\right[\cup \left]\mathrm{e}, \infty\right[$.

```
> ``=simplify(rhs((89))) assuming x>exp(1);
```

$$= \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}} \tag{91}$$

```
> ``=simplify(rhs((89))) assuming x>0, x<1/exp(1);
```

$$= \frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}} \tag{92}$$

Donc, chaque simplification correspond bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{1}{x\ln(x)\sqrt{\ln(x)^2-1}} \tag{93}$$

REMARQUE IMPORTANTE

Le choix retenu par Maple pour θ afin de définir la fonction réciproque de $\sec(\theta)$ est

$$\left[0, \frac{\pi}{2}\right[\cup \left]\frac{\pi}{2}, \pi\right] \quad \text{(Choix 1)}$$

En conséquence, pour Maple, la dérivée de $\operatorname{arcsec}(x)$ est $\dfrac{1}{|x|\sqrt{x^2-1}}$.

Dans certain manuel de mathématique on retient plutôt

$$\left[0, \frac{\pi}{2}\right[\cup \left]\pi, \frac{3\pi}{2}\right] \quad \text{(Choix 2)}$$

ce qui amène la formule $\dfrac{1}{x\sqrt{x^2-1}}$ pour la dérivée de arcsec(x).

Chacun de ces choix est arbitraire. Le « choix 1 » a comme conséquence de faire apparaître une valeur absolue dans la dérivée de arcsec(x). Le « choix 2 » est peut-être plus simple dans un premier temps mais rend fausse l'identité $\operatorname{arcsec}(x) = \arccos\left(\dfrac{1}{x}\right)$.

Selon le « choix 1 » retenu par Maple, notons que la dérivée de arcsec(x) n'est pas formulée explicitement avec une valeur absolue mais implicitement.

```
> Diff(arcsec(x),x)=diff(arcsec(x),x);
```

$$\frac{d}{dx}\operatorname{arcsec}(x) = \frac{1}{x^2\sqrt{1-\frac{1}{x^2}}} \qquad \textbf{(94)}$$

Nous allons montrer facilement que $\dfrac{1}{x^2\sqrt{1-\frac{1}{x^2}}} = \dfrac{1}{|x|\sqrt{x^2-1}}$.

```
> Équation:=1/(x^2*sqrt(1 - 1/x^2)) = 1/(abs(x)*sqrt(x^2 - 1));
```

$$\mathit{Équation} := \frac{1}{x^2\sqrt{1-\frac{1}{x^2}}} = \frac{1}{|x|\sqrt{x^2-1}} \qquad \textbf{(95)}$$

```
> simplify(Équation) assuming x<-1;
```

$$-\frac{1}{x\sqrt{x^2-1}} = -\frac{1}{x\sqrt{x^2-1}} \qquad \textbf{(96)}$$

```
> simplify(Équation) assuming x>1;
```

$$\frac{1}{x\sqrt{x^2-1}} = \frac{1}{x\sqrt{x^2-1}} \qquad \textbf{(97)}$$

Exemple 6

Calculons l'intégrale indéfinie $\displaystyle\int \frac{e^x}{4\,e^{2x}+9}\,dx$.

D'abord directement avec Maple.

```
> f:=x->exp(x)/(4*exp(2*x)+9);
```

$$f := x \to \frac{e^x}{4\,e^{2x}+9} \quad \textbf{(98)}$$

Posons l'intégrale indéfinie $\int \frac{e^x}{4\,e^{2x}+9}\,dx$ qu'il faut obtenir.

```
> Problème:=Int(f(x),x);
```

$$Problème := \int \frac{e^x}{4\,e^{2x}+9}\,dx \quad \textbf{(99)}$$

Obtenons directement cette intégrale indéfinie avec la macro-commande `value`.

```
> primitive:=value(Problème);
```

$$primitive := \frac{1}{6}\arctan\left(\frac{2}{3}\,e^x\right) \quad \textbf{(100)}$$

```
> Rép_Maple:=Problème=primitive+C;
```

$$Rép\_Maple := \int \frac{e^x}{4\,e^{2x}+9}\,dx = \frac{1}{6}\arctan\left(\frac{2}{3}\,e^x\right) + C \quad \textbf{(101)}$$

Intégrons maintenant en effectuant un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie.

```
> Problème;
```

$$\int \frac{e^x}{4\,e^{2x}+9}\,dx \quad \textbf{(102)}$$

Posons $u = 4\,e^x$.

```
> Rule[change,u=4*exp(x)]((102));
Creating problem #6

Applying substitution x = -2*ln(2)+ln(u), u = 4*exp(x) with dx = 1/u*du,
du = 4*exp(x)*dx
```

$$\int \frac{e^x}{4\,e^{2x}+9}\,dx = \int \frac{1}{4\left(4\,e^{-4\ln(2)+2\ln(u)}+9\right)}\,du \quad \textbf{(103)}$$

```
> simplify( (103) );
```

$$\int \frac{e^x}{4\,e^{2x}+9}\,dx = \int \frac{1}{u^2+36}\,du \quad \textbf{(104)}$$

```
> Rule[change,u=6*tan(theta)]((104));
Applying substitution u = 6*tan(theta), theta = arctan(1/6*u) with du =
(6+6*tan(theta)^2)*dtheta, dtheta = 1/6/(1/36*u^2+1)*du
```

$$\int \frac{e^x}{4\,e^{2x}+9}\,dx = \int \frac{1}{6}\,d\theta \quad \textbf{(105)}$$

```
> Rule[constant]((105));
```

$$\int \frac{e^{x}}{4\,e^{2x}+9}\,dx = \frac{1}{6}\,\theta \tag{106}$$

```
> Rule[revert]((106));
Reverting substitution using theta = arctan(1/6*u)
```

$$\int \frac{e^{x}}{4\,e^{2x}+9}\,dx = \frac{1}{6}\arctan\left(\frac{1}{6}\,u\right) \tag{107}$$

```
> Rule[revert]((107));
Reverting substitution using u = 4*exp(x)
```

$$\int \frac{e^{x}}{4\,e^{2x}+9}\,dx = \frac{1}{6}\arctan\left(\frac{2}{3}\,e^{x}\right) \tag{108}$$

```
> Rép_finale:=(108)+(0=C);
```

$$Rép\_finale := \int \frac{e^{x}}{4\,e^{2x}+9}\,dx = \frac{1}{6}\arctan\left(\frac{2}{3}\,e^{x}\right) + C \tag{109}$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x}\left(\frac{1}{6}\arctan\left(\frac{2}{3}\,e^{x}\right) + C\right) = \frac{1}{9}\,\frac{e^{x}}{\frac{4}{9}\left(e^{x}\right)^{2}+1} \tag{110}$$

```
> ``=simplify(rhs((110)));
```

$$= \frac{e^{x}}{4\,e^{2x}+9} \tag{111}$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{e^{x}}{4\,e^{2x}+9} \tag{112}$$